



Lawrence Berkeley Laboratory

UNIVERSITY OF CALIFORNIA

Accelerator & Fusion Research Division

RECEIVED
LAWRENCE
BERKELEY LABORATORY

OCT 13 1989


LIBRARY AND
DOCUMENTS SECTION

A Mathematica Notebook of Relativistic Kinematics

Application to B Factories, Elastic Scattering and Bremsstrahlung

B. Autin

July 1989



LBL-27535
-2

DISCLAIMER

This document was prepared as an account of work sponsored by the United States Government. While this document is believed to contain correct information, neither the United States Government nor any agency thereof, nor the Regents of the University of California, nor any of their employees, makes any warranty, express or implied, or assumes any legal responsibility for the accuracy, completeness, or usefulness of any information, apparatus, product, or process disclosed, or represents that its use would not infringe privately owned rights. Reference herein to any specific commercial product, process, or service by its trade name, trademark, manufacturer, or otherwise, does not necessarily constitute or imply its endorsement, recommendation, or favoring by the United States Government or any agency thereof, or the Regents of the University of California. The views and opinions of authors expressed herein do not necessarily state or reflect those of the United States Government or any agency thereof or the Regents of the University of California.

**A *Mathematica* Notebook of Relativistic Kinematics.
Application to B factories, elastic scattering and Bremsstrahlung.***

Bruno Autin [†]

**Accelerator & Fusion Research Division
Lawrence Berkeley Laboratory
University of California
1 Cyclotron Road
Berkeley, California 94720**

July 1989

* Work performed under the auspices of the U.S. Department of Energy by the Lawrence Berkeley Laboratory under contract No. DE-AC03-76SF00098.

[†] Permanent address: CERN, PS Division, 1211 Geneva 23, Switzerland.

Contents

	page
1. Introduction	1
2. Reference frames	2
3. Beam energy	
3.1 Unequal energies	3
3.2 Equal energies	5
3.3 Special cases	6
4. Center of mass momentum	
4.1 Unequal energies	7
4.2 Equal energies	8
4.3 Special cases	8
5. Lorentz transformations	
5.1 Matrix form	10
5.2 Product of two Lorentz transformations	10
5.3 Successive transformations of a 4-vector	11
5.4 Time component of a 4-vector	11
5.5 Space components of a 4-vector	11
6. B-factory	
6.1 Symmetric collisions at a finite crossing angle	12
6.2 Asymmetric head-on collisions	15
7. Elastic scattering cross-section	
7.1 Electron positron type collisions	21
7.2 Electron electron type collisions	23
7.3 Coulomb scattering	24
8. Bremsstrahlung cross-section	26
References	27
Index	28

■ 1. Introduction

Mathematica [1] has raised recently a great interest in the scientific community for its symbolic and graphic facilities. In the context of this notebook, its use in the field of relativistic kinematics has been motivated by studies on B-factories where the variety of possible configurations for the colliding beams requires a systematic general approach.

The first part of this paper (sections 2, 3, 4, 5) deals with the coding of general expressions [2] and graphical illustrations. Taking the energy \sqrt{s} available in the center of mass of the collision as a scaling parameter, the energy of one beam as a function of the energy of the other beam and the characteristics of the center of mass are derived. The matrix of a Lorentz transformation and the transformation of a 4-vector through several different reference frames are then established. The various commands are listed in an Index at the end of the note. B-factories, elastic scattering and Bremsstrahlung are studied in the second part (sections 6, 7, 8).

The notebook has been written in the *Mac Intosh* version of *Mathematica* and the reader who is not familiar with this version should browse the *User Manual* and especially the section **Cells**. Initialization must be performed at the opening. The symbol # denotes the activation of the cell which follows the instruction.

■ 2. Reference frames

All the frames are deduced from each other by translation. The horizontal plane (x, z) is defined by the two colliding beams and, for the center of mass system, ψ is equal to zero. The z -axis in the laboratory frame is the bisectrix of the beam axes.




Figure 1. Reference frames

In the particle frame, decays of the particle may be observed (section 6.2).

■ 3. Beam energy

□ 3.1 Unequal energies

Let us consider two beams colliding with an angle 2θ . The state of a particle in a given beam is defined by an energy - momentum vector (E, p) whose norm is the square of the mass of the particle

$$E^2 - p^2 = m^2 \quad (1)$$

All the energies and masses are normalized to the energy \sqrt{s} available in the center of mass. The center of mass energy-momentum vector is

$$[E, p]_{cm} = [E_1 + E_2, (p_1 - p_2) \cos \theta, 0, (p_1 + p_2) \sin \theta] \quad (2)$$

so that the momentum modulus is

$$p_{cm} = \sqrt{p_1^2 + p_2^2 - 2 p_1 p_2 \cos 2\theta} \quad (3)$$

The expression of E_2 as a function of E_1 is the solution of (1) applied to the center of mass

$$m'^2 - 2 E_1 E_2 = 2 \sqrt{(E_1^2 - m_1^2)(E_2^2 - m_2^2)} \cos 2\theta \quad (4)$$

with

$$m'^2 = 1 - m_1^2 - m_2^2 \quad (5)$$

The equation (4) is solved by squaring the two sides. The sign ambiguity which is thus created is eliminated by imposing that the general solution is reduced to

$$E_2 = \frac{1}{2 E_1 (1 + \cos 2\theta)} \quad (6)$$

in the limit case where the masses are zero.

Relativistic Kinematics

(* general purpose tools *)

(* solver for 2nd degree equation with 1 physical solution *)

<<Solve2.m

(* trigonometry rules *)

r1 =a_ Cos[b_]^2+a_ Sin[b_]^2->a

r2 =(Cos[b_]-Sin[b_])*(Cos[b_]+Sin[b_])->Cos[2 b]

r12=Sin[b_]^2->1-Cos[b]^2

(* definitions of m' and p1 *)

momentum[e_,m_] :=Sqrt[e^2-m^2]

mprime[m1_,m2_] :=Sqrt[1-m1^2-m2^2]

r3 =e1^2-p1^2+m2^2-1->-m'^2

r3I=m' ->Sqrt[-m1^2-m2^2+1]

r4 =p1->Sqrt[e1^2-m1^2]

r5=Abs[p1 Cos[2 t]]->p1 Cos[2 t]; (* Rule for sign definition *)

(* Equation relating beam 2 energy e2 to beam 1 energy e1 *)

eq12[e1_,e2_,t_,m1_,m2_]=

Block[{a1,a2,p1,p2,pS,lhs,rhs},

pS=Expand[(p1-p2)^2 Cos[t]^2+(p1+p2)^2 Sin[t]^2]//.r1;

a1=Select[pS,!FreeQ[#,p1 p2 _]&];

a2=pS-a1;

a1=Factor[a1]//.r2;

p2=Sqrt[e2^2-m2^2];

lhs=a1^2;

rhs=(Expand[(e1+e2)^2-a2-1]//.r3)^2;

lhs-rhs (* Equation (4) squared *)]

(* Calculation of e2 *)

f[e1_,t_,m1_,m2_]=

Block[{a1,a2,e2,p1,p2,pS,lhs,rhs,den,num,u},

u=e2/.Solve2[eq12[e1,e2,t,m1,m2]==0,e2,False];

den=Numerator[1/u];

num=Denominator[1/u];

a1=Select[den,!FreeQ[#,Sqrt[_]]&];

a2=den-a1;

den=a2-Sqrt[Factor[a1^2]]//.r5 ;

Factor[num/den] /.r4]

The expression of E2 is thus in a reduced form

$$E_2 = \frac{(2 m_2 p_1 \cos 2\theta)^2 + m^4}{2 (e_1 m^2 + p_1 \cos 2\theta \sqrt{(2 m_2 p_1 \cos 2\theta)^2 + m^4 - 4 e_1^2 m_2^2})} \quad (7)$$

which is always well conditioned for any value of the variables. In practical calculations, it is necessary to have m' in an expanded form, this is the purpose of the the following cell.

```
energy2[e1_,t_,m1_,m2_]=f[e1,t,m1,m2]/.r3I
```

□ 3.2 Equal energies

We shall now derive the analytic expression of the beam energy for a given value of \sqrt{s} when the two beams have the same energy by solving

$$E_2 = E_1 \quad (8)$$

(* Calculation of the square of the beam energy eS *)

```
energyS[t_,m1_,m2_]=Block[{a1,a2,num,den,eq,e,u,v},
  num=Numerator[Together[f[e,t,m1,m2]-e]];
  a1=Select[num,!FreeQ[#,Sqrt[_]]&];
  a2=num-a1;
  a1=Collect[Expand[a1^2],e];
  a2=Collect[Expand[a2^2],e];
  (* eq is of 3-rd degree in e^2 before factorization *)
  eq=Factor[Collect[a1-a2,e]];
  (* eq is of 2-nd degree in e^2 after factorization *)
  eq=Select[eq,!FreeQ[#,e^4]&];
  eq=Collect[eq,e]//.e->Sqrt[u];
  v=u/.Solve2[eq==0,u,False];
  den=Numerator[1/v];
  num=Denominator[1/v];
  a1=Select[den,!FreeQ[#,Sqrt[_]]&];
  a2=den-a1;
  a1=Factor[CoefficientList[a1^2,Cos[2 t]]/16].
    {0,0,1,0,Cos[2 t]^2} Cos[2 t]^2 16;
  a1=Sqrt[a1];
  a2=Factor[CoefficientList[a2,Cos[2 t]]/4].
    {1,0,Cos[2 t]^2} 4;
  v=(num/2)/(a1/2+a2/2)/.r3I/.Abs[Cos[2 t]]->Cos[2 t]]
```

or, with more compact notations,

$$E^2 = \frac{-(2 m_2 m_1 \cos 2\theta)^2 + m^4}{2 (m^2 - (m_1^2 + m_2^2) \cos^2 2\theta + \cos 2\theta \sqrt{((m_1^2 - m_2^2) \cos 2\theta)^2 + (m^2 - 2 m_1^2)(m^2 - 2 m_2^2)})} \quad (9)$$

□ 3.3 Special cases

The expressions (6) and (9) become simpler in special cases whose most usual are listed in the following cells. # # # #

(* Special case 1: equal masses *)

energy2[e1, t, m, m]

$$E_2 = \frac{(1 - 2m^2)^2 + (2mp_1 \cos 2\theta)^2}{2(E_1(1 - 2m^2) + p_1 \cos 2\theta \sqrt{(1 - 2m^2)^2 - 4m^2(E_1^2 - p_1^2 \cos^2 2\theta)})} \quad (10)$$

energyS[t, m, m]

$$E^2 = \frac{(1 - 2m^2)^2 + (2m^2 \cos 2\theta)^2}{2(1 - 2m^2(1 + \cos^2 2\theta) + \cos 2\theta(1 - 4m^2))} \quad (11)$$

(* Special case 2: zero crossing angle and equal masses *)

energy2[e1, 0, m, m]

$$E_2 = \frac{(1 - 2m^2)^2 + (2mp_1)^2}{2(E_1(1 - 2m^2) + p_1 \sqrt{1 - 4m^2})} \quad (12)$$

r5=1-8 a_^2+16 a_^4->(1-4 a^2)^2

Factor[energyS[0, m, m]/.r5]

$$E = \frac{1}{2} \quad (13)$$

(* Special case 3: negligible masses and finite crossing angle *)

Factor[energy2[e1, t, 0, 0]]

$$E_2 = \frac{1}{4E_1 \cos^2 \theta} \quad (14)$$

Factor[energyS[t, 0, 0]]

$$E = \frac{1}{2 \cos \theta} \quad (15)$$

(* Special case 4: negligible masses and zero crossing angle *)

energy2[e1, 0, 0, 0]

$$E_2 = \frac{1}{4E_1} \quad (16)$$

■ 4. Center of mass momentum

□ 4.1 Unequal energies

The expression of the center of mass momentum

$$p_{cm} = \sqrt{1 - (E_1 + E_2)^2} \quad (17)$$

is, in principle, fully defined once E_1 and E_2 are known. However, instead of substituting E_2 directly, it is much more efficient to first substitute the square of E_2 , drawn from equation (4) then E_2 itself. In other words, p_{cm} is the square root of a linear function of E_2 . Nevertheless the full expression is very complicated and we leave it in *Mathematica* language.

(* Modulus of p_{cm} *)

```
e2S[e1_,t_,m1_,m2_] := Block[{clist, e2, e2S, n},
  rr = Abs[e1] -> e1;
  m' = mprime[m1, m2];
  p1 = momentum[e1, m1];
  eq = eq12[e1, e2, t, m1, m2] /. rr;
  n = Exponent[Expand[eq], e2];
  clist = CoefficientList[eq, e2];
  (* The following condition prevents ill conditioning of e2S *)
  If[n == 1,
    (clist[[1]]/clist[[2]])^2,
    Expand[-clist[[1]] - clist[[2]] e2 / clist[[3]] ] ]
pcmMa[e1_,t_,m1_,m2_] :=
  Sqrt[e2S[e1,t,m1,m2] + 2 e1 e2 + e1^2 - 1] /.
  e2 -> energy2[e1,t,m1,m2]
```

The center of mass momentum vector is calculated and plotted in relation with the momentum vectors of the primary beams in the next two cells.

(* Components of p_1 , p_2 , p_{cm} *)

```
p1Va [e1_,t_,m1_,m2_] := { Cos[t], Sin[t] } momentum[e1,m1]
p2Va [e1_,t_,m1_,m2_] := { -Cos[t], Sin[t] } *
  Sqrt[e2S[e1,t,m1,m2] - m2^2] /. e2 -> energy2[e1,t,m1,m2] /. rr
pcmVa[e1_,t_,m1_,m2_] := p1Va[e1,t,m1,m2] + p2Va[e1,t,m1,m2]
```

(* Graph of p1, p2, pcm *)

```
graphVa[e1_,t_,m1_,m2_] :=
Show[Graphics[
  {Line[{{0,0},pcmVa[e1,t,m1,m2]}],Text["pcm",pcmVa[e1,t,m1,m2],{0,
    Line[{{0,0},p1Va[e1,t,m1,m2]}],Text["p1",p1Va[e1,t,m1,m2],{0,
    Line[{{0,0},p2Va[e1,t,m1,m2]}],Text["p2",p2Va[e1,t,m1,m2],{0,
    PointSize[.02],Point[{0,0}],
      Point[pcmVa[e1,t,m1,m2]],
      Point[p1Va[e1,t,m1,m2]],
      Point[p2Va[e1,t,m1,m2]]}]]]
```

□ 4.2 Equal energies

The rules given in section 3.2 for the energy are resumed for the momentum in the next cell with θ as the independent variable.

(* Modulus of pcm *)

```
pcmMs[t_,m1_,m2_] := Sqrt[Abs[1-4 energyS[t,m1,m2]]]
```

(* Components of p1, p2, pcm *)

```
p1Vs[t_,m1_,m2_] := {Cos[t],Sin[t]} Sqrt[energyS[t,m1,m2]-m1^2]
p2Vs[t_,m1_,m2_] := {-Cos[t],Sin[t]} Sqrt[energyS[t,m1,m2]-m2^2]
pcmVs[t_,m1_,m2_] := p1Vs[t,m1,m2]+p2Vs[t,m1,m2]
```

(* Graph of p1, p2, pcm *)

```
graphVs[t_,m1_,m2_] :=
Show[Graphics[
  {Line[{{0,0},pcmVs[t,m1,m2]}],Text["pcm",pcmVs[t,m1,m2],{0,1}],
    Line[{{0,0},p1Vs[t,m1,m2]}],Text["p1",p1Vs[t,m1,m2],{0,1}],
    Line[{{0,0},p2Vs[t,m1,m2]}],Text["p2",p2Vs[t,m1,m2],{0,1}],
    PointSize[.02],Point[{0,0}],
      Point[pcmVs[t,m1,m2]],
      Point[p1Vs[t,m1,m2]],
      Point[p2Vs[t,m1,m2]]}]]]
```

□ 4.3 Special cases

The center of mass momentum can be evaluated in the special cases listed in section 3.3. It is left as an exercise to code the simplifying rules. ###

Relativistic Kinematics

(* Special case 1: equal masses *)

`pcmMa[e1,t,m,m]`

`pcmMs[t,m,m]`

(* Special case 2: equal masses *)

`pcmMa[e1,t,m,m]`

`pcmMs[0,m,m]`

(* Special case 3: negligible masses and finite crossing angle *)

`pcmMa[e1,t,0,0]`

`pcmMs[t,0,0]`

(* Special case 4: negligible masses and zero crossing angle *)

`pcmMa[e1,0,0,0]`

5. Lorentz transformations

□ 5.1 Matrix form

A Lorentz transformation from a moving frame to an observer frame is completely defined by the *rapidity* ζ or, equivalently, by the product $\beta\gamma$ of the relativistic parameters of the moving frame

$$\zeta = \sinh^{-1} \beta\gamma \quad (18)$$

and by the unit vector \mathbf{u} of the direction of motion

$$u_x = \cos \psi \sin \phi \quad (19)$$

$$u_y = \sin \psi \quad (20)$$

$$u_z = \cos \psi \cos \phi \quad (21)$$

If a particle of mass M and momentum p has been created in a reference frame R , the rapidity of the particle reference frame with respect to R is given by

$$\zeta = \sinh^{-1} \frac{p}{M} \quad (22)$$

(* Matrix of a Lorentz transformation *)

```
Lorentz[bg_,phi_,psi_] := Block[{cz,sz},
  cz=Sqrt[1+bg^2];
  sz=bg;
  ux=Cos[psi] Sin[phi];
  uy=Sin[psi];
  uz=Cos[psi] Cos[phi];
  { {cz, sz ux, sz uy, sz uz},
    {sz ux, 1+(cz-1) ux^2, (cz-1) ux uy, (cz-1) ux uz},
    {sz uy, (cz-1) ux uy, 1+(cz-1) uy^2, (cz-1) uy uz},
    {sz uz, (cz-1) ux uz, (cz-1) uy uz, 1+(cz-1) uz^2} }]
```

□ 5.2 Product of two Lorentz transformations

The product of two Lorentz transformations is a Lorentz transformation. The property of symmetry is used for the construction of the product matrix.

(* Product of two Lorentz transformations *)

```
LorentzProd[lor2_,lor1_]:=Block[{a},
a=Table[If[j<i,0,lor2[[i]].lor1[[j]],{i,4},{j,4}];
Table[If[j<i,a[[j,i]],a[[i,j]],{i,4},{j,4}]]
```

□ 5.3 Successive transformations of a 4-vector

The 4-vector may be a space-time vector or an energy-momentum vector. If there are n transformations, the final vector \mathbf{v}_f must be declared as a function of the Lorentz parameters $(\beta_j, \phi_j, \psi_j)$, and of the initial vector $\mathbf{v}_i(t, x, y, z)$ in the following way

$$\text{LorentzV}[\{(\beta\gamma)_1, \phi_1, \psi_1\}, \dots, \{(\beta\gamma)_n, \phi_n, \psi_n\}], \{t, x, y, z\}$$

```
LorentzV[lorPar_,vectorI_]:=Block[{n,timespace},
n=Length[lorPar];
timespace=vectorI;
Do[timespace=
Lorentz[lorPar[[i,1]],lorPar[[i,2]],lorPar[[i,3]]].
timespace,{i,n}];
timespace]
```

□ 5.4 Time component of a 4-vector

The time component is the first component of the 4-vector

```
time[vector_]:=vector[[1]]
```

As the norm of a space-time vector is zero, the time component is also the length of the space vector. The norm of the energy-momentum vector has already been discussed in sections 3 and 4.

□ 5.5 Space components of a 4-vector

The space components are the last three components of the 4-vector.

```
space[vector_]:=Take[vector,-3]
```

We have just seen that it is useless to calculate the norm of the space vector but one may wish to compare the transformed vector with the initial vector.

(* Graph of $\mathbf{v}_1, \mathbf{v}_2$ *)

```
graphS[v1_,v2_]:=
Show[Graphics3D[{Line[{{0,0,0},v1}],Line[{{0,0,0},v2}]}]]
```

6. B-factory

The production of B-mesons in this type of machine results from the decay of the Υ ($4s$) resonance into a particle-antiparticle pair [3]:

$$\Upsilon \rightarrow B + \bar{B}$$

The particle data are drawn from [4]. The Υ mass is $\sqrt{s} = 10.58$ GeV. The B mass is 5.277 GeV and thus close to its total energy $\sqrt{s}/2$ so that its momentum and its rapidity are small:

$$p_B = \sqrt{\frac{s}{4} - M_B^2} = .371 \text{ GeV}/c$$

$$(\beta\gamma)_B = \frac{p_B}{M_B} = .07$$

Its proper life-time is $\tau = 1.3$ ps. If the center of mass of the colliding beams is at rest, both momentum and life-time are too short to differentiate the decays between the particle and the antiparticle. Two methods have been contemplated to boost the center of mass and therefore dilate the flight path in the laboratory: collision of beams of equal energies at large angle [5] and head-on collisions of beams of unequal energies [3].

□ 6.1 Symmetric collisions at a finite crossing angle

We are in the conditions of section 3.2. The beam energy which is necessary to produce the Υ -resonance is given by (14) and the variation of the energy normalized to \sqrt{s} with the half crossing angle θ is programmed in the next cell. #

(* Graph 1: energy versus t *)

```
Plot[Sqrt[energyS[t, 0, 0]], {t, 0, Pi/4}, AxesLabel->{"t", "e"}]
```




Figure 2. Variation of the beam energy normalized to \sqrt{s} versus the half crossing angle

Relativistic Kinematics

The energy must be increased with respect to its minimum value $1/2$ by the factor $1 / \cos \theta$. The of the momentum can also be found to vary like $\tan \theta$. #

(* Graph 2: center of mass momentum versus t *)

```
Plot [pcmMs[t, 0, 0], {t, 0, Pi/4}, AxesLabel->{"t", "p"}]
```




Figure 3. Variation of the center of mass momentum normalized to \sqrt{s} versus the half crossing angle

For a given value of θ , say $\pi/4$, the configuration of the momentum vectors and the numerical values with, for instance 10 decimal places, can be obtained. ###

(* Graph 3: momentum vectors *)

```
graphVs [Pi/4, 0, 0]
```




Figure 4. Beams and center of mass momentum vectors

(* Numerical value of the normalized energy *)

`N[Sqrt[energyS[Pi/4, 0, 0]], 10]`

(* Numerical value of the normalized momentum *)

`N[Sqrt[pcmMs[Pi/4, 0, 0]], 10]`

Let us see now how the particle space-time vector (1, 0, 1, 0) is transformed from the center of mass frame to the laboratory frame. The parameters of the Lorentz transformation are

$$\{ (\beta\gamma), \phi, \psi \}_{cm} = \{ p_{cm}, \pi/2, 0 \} \quad (23)$$

We define **v1**, calculate **v2**, represent the space part of **v1** and **v2** for a special value of θ , re-calculate **v2** for any θ and plot the variation of the flight path as a function of θ . ##

(* Graph 4: space-time vectors *)

`v1={1, 0, 1, 0}`

`v2=LorentzV[{{pcmMs[Pi/4, 0, 0], Pi/2, 0}}, v1]`

`graphS[space[v1], space[v2]]`




Figure 5. Space-time vectors before and after Lorentz transformation

(* Graph 5: flight path length versus t *)

```
v2=LorentzV[{{pcmMs[t,0,0],Pi/2,0}},v1]
Plot[time[v2],{t,0,Pi/4},AxesLabel->{"t","1/tau"}]
```




Figure 6. Variation of flight path length normalized to the B life time versus θ

□ 6.2 Asymmetric head-on collisions

The conditions are now those of section 3.1 with zero crossing angle and negligible masses. The numerical value of $E2/\sqrt{s}$ is obtained by substituting the value of $E1/\sqrt{s}$, θ , $m1$, $m2$ into the arguments of the function **energy2**. #

(* Numerical value of $e2/\sqrt{s}$ *)

```
N[energy2[3/10.58,0,0,0],10]
```

The graphs $E2$ versus $E1$ and $E2$ versus $(E1,\theta)$ can be plotted. ##

```
Plot[energy2[e1,0,0,0],{e1,.1,5},AxesLabel->{"e1","e2"}]
```

Relativistic Kinematics




Figure 7. Variation of beam 2 energy versus beam 1 energy.
Both energies are normalized to \sqrt{s} .

```
Plot3D[energy2[e1, t, 0, 0], {e1, .1, 5}, {t, 0, Pi/6}, PlotRange->All]
```




Figure 8. Variation of beam 2 energy versus beam 1 energy and half crossing angle

For the study of the center of mass, we follow the same steps as in section 6.1. #####

(* Numerical value of pcm *)

```
N[pcmMa[5/10.58, 0, 0, 0], 10]
```

Relativistic Kinematics

(* Graph 8: center of mass momentum versus t *)

```
y=pcmMa[e1,0,0,0]
Plot[y,{e1,0.1,5},AxesLabel->{"e1","p"}]
```




Figure 9. Variation of center of mass momentum versus beam 1 energy . For the low values of e_1 , the center of mass moves in the direction of beam 2; when the beams have the same energy, it is at rest; then, for the high values of e_1 , it moves in the direction of beam 1.

```
N[pcmVa[3/10.58,0,0,0],10]
```

(* Graph 9: momentum vectors *)

```
graphVa[3/10.58,0,0,0]
```



Figure 10. Momentum vectors in head-on collisions.

The parameters of the Lorentz transformation from the center of mass frame to the laboratory frame are

$$\{(\beta\gamma), \phi, \psi\}_{cm} = \{p_{cm}, 0, 0\} \quad (24)$$

###

(* Graph 10: space-time vectors *)

`v1={1, 0, 1, 0}`

`v2=LorentzV[{{pcmMa[3/10.58, 0, 0, 0]}, 0, 0}], v1]`

`graphS[space[v1], space[v2]]`




Figure 11. Space-time vectors in head-on collisions.

(* Graph 11: flight path length *)

`length=time[LorentzV[{{pcmMa[e1, 0, 0, 0]}, 0, 0}], v1]`

`Plot[length, {e1, .1, 5}, AxesLabel->{"e1/√s", "1/tau"}]`




Figure 12. Variation of flight path length normalized to the B life time versus $e1/\sqrt{s}$

As the B particle has a small transverse momentum, its direction of motion is almost colinear to the beam axis in the laboratory frame and its vertex detection is inaccurate. For this reason, it is not observed directly but through a secondary decay like J/ψ K_s for which the momentum is 1.8 GeV in the B frame. We can compare the orientation of the B and K_s momentum vectors in the laboratory frame. For the B particle, the initial momentum vector is $\{ 1/2, .035, 0, 0 \}$ and it is subject to the Lorentz transformation (24). For the K_s particle, the initial momentum vector is $\{ .1764, .17, 0, 0 \}$ and it is subject to the product of Lorentz transformations (24) and (25):

$$\{ (\beta\gamma), \phi, \psi \}_B = \{ .07, \pi/2, 0 \} \quad (25)$$

#

(* Graph 12: momentum vectors *)

```
v1i={.5, .035, 0, 0}
v1f=LorentzV[{{pcmMa[3/10.58, 0, 0, 0], 0, 0}}, v1i]
v2i={.1764, .17, 0, 0}
v2f=LorentzV[{{.07, Pi/2, 0}, {pcmMa[3/10.58, 0, 0, 0], 0, 0}}, v2i]
graphS[space[v1f], space[v2f]]
```

Relativistic Kinematics




Figure 13. B and K_s momentum vectors in the laboratory frame. Both are parallel to the x-z plane but the B momentum vector lies along the z-axis whereas the K_s momentum vector makes a 58 degree angle with the z-axis.

■ 7. Elastic scattering cross-section

When a charged particle of momentum p_1 enters the Coulomb field of another particle at some impact parameter, it is deflected by an angle χ . The differential cross-section of the interaction of the two particles is a function of χ which depends on the various processes involved in the interaction. We use the expressions which have been derived for electron - positron, electron-electron and electron-fixed field interactions. However, the same expressions can be applied to any type of particles when the field contribution, represented by the coupling constant α , is clearly distinguished from the particle masses; for this reason, the classical radius of the electron is never used. The total cross-sections are expressed as functions of a minimum deflection angle χ_m . In the context of particle colliders, one is interested in the beam life time and therefore in particles scattered outside the machine acceptance, χ_m is then the beam angular divergence where the interaction takes place.

□ 7.1 Electron positron type collisions

The differential cross section of electron-positron or *Bhabha* scattering for ultra-relativistic particles is [6]

$$d\sigma_{\text{Bhabha}} = \frac{(\alpha h c / 2\pi)^2}{2 s} \left[\frac{1 + \cos^4 \frac{\chi}{2}}{\sin^4 \frac{\chi}{2}} - 2 \frac{\cos^4 \frac{\chi}{2}}{\sin^2 \frac{\chi}{2}} + \frac{1 + \cos^2 \chi}{2} \right] d\Omega \quad (26)$$

where α is the fine structure constant, h the Planck constant, c the light velocity, χ the scattering angle in the center of mass frame and $d\Omega$ the differential solid angle $2\pi d(-\cos \chi)$. The total cross-section is obtained by integration of the differential cross-section from a minimum angle χ_m to π . In the use of the code, \sqrt{s} must be expressed in GeV and the cross-sections are given in mbarn [10^{-27} cm^2].

Relativistic Kinematics

(* Bhabha cross-section in the center of mass frame [mb] *)

(* more rules *)

r6=Cos[t_/2]->Sqrt[(1+Cos[t])/2]

r7=Sin[t_/2]->Sqrt[(1-Cos[t])/2]

r8=Cos[t]->x

Unprotect[Log]

r9=Log[a_]-Log[b_]->Log[a/b]

Protect[Log]

Bhabha[xm_, s_] = Block[{f, g, h, h1, h2, t, x},

(* differential cross-section as a function of x = cos t *)

f=(1+Cos[t/2]^4)/Sin[t/2]^4-2 Cos[t/2]^4 /Sin[t/2]^2 +
(1+Cos[t]^2)/2 // .r6//.r7//.r8;

(* total cross-section *)

g=Integrate[f, {x, -1, xm}];

h=Expand[g];

h1=Select[h, !FreeQ[#, Log[_]]&];

h2=Select[h, FreeQ[#, Log[_]]&];

h1=Factor[h1]/.r9;

h2=Factor[h2];

(h1+h2) .389/(137)^2 Pi/s]

or

$$\sigma_{\text{Bhabha}} = \frac{\pi (\alpha h c / 2\pi)^2}{s} \left[\frac{(1 + \cos \chi_m) (49 - 23 \cos \chi_m - \cos^2 \chi_m - \cos^3 \chi_m)}{6 (1 - \cos \chi_m)} + 8 \text{Log} \frac{1 - \cos \chi_m}{2} \right] \quad (27)$$

As χ_m is a small angle, we expand σ for values of $x_m = \cos \chi_m$ near 1.

Bhabha1[xm_, s_]=Normal[Series[Bhabha[xm, s], {xm, 1, -1}]]

The minimum angle is actually known in the laboratory frame. In the case of asymmetric head-on collisions, it is determined in the center of mass frame by applying to the vector

$$\{ E, 0, 0, \sqrt{E^2 - m^2} \cos \chi_0 \}$$

the Lorentz transformation inverse of (24). m is the mass of the scattered particle and E its energy. $\cos \chi_m$ is then given by the ratio of the z -component to the momentum of the transformed vector.

```
(* asymmetric head-on collisions *)
Bhabhaa[e_,m_,s_,chi0_]:=Block[{v1,v2,xm},
v1={e,0,0,Sqrt[e^2-m^2]Cos[chi0]};
v2=LorentzV[{-pcmMa[e,0,m,m],0,0},v1];
xm=v2[[4]]/Sqrt[(v2[[1]]-m)(v2[[1]]+m)];
Bhabha1[xm,s]
]
```

For a symmetric collision at finite angle, the situation is a little more complicated. One has to apply the inverse of (23) to the beam momentum vector

$$\{ E, \sqrt{E^2 - m^2} \sin \theta, 0, \sqrt{E^2 - m^2} \cos \theta \}$$

and to the scattered particle momentum vector

$$\{ E, \sqrt{E^2 - m^2} \cos \chi_0 \sin \theta, \sqrt{E^2 - m^2} \sin \chi_0, \sqrt{E^2 - m^2} \cos \chi_0 \cos \theta \}$$

The cosine of the scattering angle in the center of mass frame is then deduced from the scalar product of the momentum parts of the transformed vectors.

```
(* finite angle symmetric collisions *)
Bhabhas[t_,m_,s_,chi0_]:=Block[{es,ps,v1,v2,vb0,vb1,vb2,xm},
es=Sqrt[energyS[t,m,m]];
ps=Sqrt[energyS[t,m,m]-m^2];
v1={es,ps Cos[chi0] Sin[t],ps Sin[chi0],ps Cos[chi0] Cos[t]};
vb0=p1Vs[t,m,m];
vb1={es,vb0[[2]],0,vb0[[1]]};
v2=LorentzV[{-pcmMs[t,m,m],Pi/2,0},v1];
vb2=LorentzV[{-pcmMs[t,m,m],Pi/2,0},vb1];
xm=space[v2].space[vb2]/
Sqrt[(v2[[1]]-m)(v2[[1]]+m)(vb2[[1]]-m)(vb2[[1]]+m)];
Bhabha1[xm,s]]
```

□ 7.2 Electron electron type collisions

The differential cross section of electron-electron or *Møller* scattering for ultra-relativistic particles is [6]

$$d\sigma_{\text{Moller}} = \frac{(\alpha h c / 2\pi)^2}{s} \left[\frac{16}{\sin^4 \chi} - \frac{8}{\sin^2 \chi} + 1 \right] d\Omega \quad (28)$$

with the notations of 7.1. As the two particles are identical, the total cross-section is obtained by integrating the differential cross-section over the χ interval $\{ 0, \pi / 2 \}$.

Relativistic Kinematics

```
(* Møller cross-section in the center of mass frame [mb] *)
```

```
(* more rules *)
```

```
r10=Sin[t_]->Sqrt[(1-Cos[t]^2)]
```

```
Moller[xm_, s_] = Block[{f, g, t, x},
```

```
(* differential cross-section as a function of x = cos t *)
```

```
f=16/Sin[t]^4-8/Sin[t]^2+1//.r10//.r8;
```

```
(* total cross-section *)
```

```
g[x_]=Expand[Integrate[f,x]];
```

```
g[xm] .389/(137)^2 2 Pi/s
```

$$\sigma_{\text{Moller}} = \frac{(\alpha h c / 2\pi)^2}{s} \left(1 + \frac{8}{\sin^2 \chi_m} \right) \cos \chi_m \quad (29)$$

The determination of the total cross-section in the various beam configurations follows the same steps as in section 7.1.

```
(* asymmetric head-on collisions *)
```

```
Mollera[e_, m_, s_, chi0_] := Block[{v1, v2, xm},
```

```
v1={e, 0, 0, Sqrt[e^2-m^2]Cos[chi0]};
```

```
v2=LorentzV[{{-pcmMa[e, 0, m, m], 0, 0}}, v1];
```

```
xm=v2[[4]]/Sqrt[(v2[[1]]-m)(v2[[1]]+m)];
```

```
Moller[xm, s]
```

```
]
```

```
(* finite angle symmetric collisions *)
```

```
Mollers[t_, m_, s_, chi0_] := Block[{es, ps, v1, v2, vb0, vb1, vb2, xm},
```

```
es=Sqrt[energyS[t, m, m]];
```

```
ps=Sqrt[energyS[t, m, m]-m^2];
```

```
v1={es, ps Cos[chi0] Sin[t], ps Sin[chi0], ps Cos[chi0] Cos[t]};
```

```
vb0=p1Vs[t, m, m];
```

```
vb1={es, vb0[[2]], 0, vb0[[1]]};
```

```
v2=LorentzV[{{-pcmMs[t, m, m], Pi/2, 0}}, v1];
```

```
vb2=LorentzV[{{-pcmMs[t, m, m], Pi/2, 0}}, vb1];
```

```
xm=space[v2].space[vb2]/
```

```
Sqrt[(v2[[1]]-m)(v2[[1]]+m)(vb2[[1]]-m)(vb2[[1]]+m)];
```

```
Moller[xm, s]
```

□ 7.3 Coulomb scattering

The differential cross section of Coulomb scattering in the target particle frame which is assumed to be the same as the laboratory frame is, for ultra-relativistic particles, [6]

$$d\sigma_{\text{Coulomb}} = \left(\frac{Z \alpha h c / 2\pi}{2 E_1} \right)^2 \frac{\cos^2 \frac{\chi}{2}}{\sin^4 \frac{\chi}{2}} d\Omega \quad (30)$$

where E_1 is the total energy of the incident particle and Z the product of the incident and target particles number of charges. The total cross-section is obtained by integrating the differential cross-section over the χ interval $(0, \pi)$.

(* Coulomb cross-section in the laboratory frame [mb] *)

```
Coulomb[xm_, e1_, z_] = Block[{alpha, f, g, h1, h2, r2, t, x},
```

```
(* differential cross-section as a function of x = cos t *)
f=Cos[t/2]^2/Sin[t/2]^4//.r6//.r7//.r8;
```

```
(* total cross-section *)
g=Integrate[f, {x, -1, xm}];
h1=Select[g, !FreeQ[#, Log[_]]&];
h2=Select[g, FreeQ[#, Log[_]]&];
h1=Factor[h1]//.r9;
alpha=1/137;
r2=.389/e1^2; (* [ mb ] *)
(h1+h2) .5 Pi (z alpha)^2 r2]
```

or

$$\sigma_{\text{Coulomb}} = \pi \left(\frac{Z \alpha h c / 2\pi}{E_1} \right)^2 \left[\frac{1}{\sin^2 \frac{\chi_m}{2}} + 2 \text{Log} \left(\sin \frac{\chi_m}{2} \right) - 1 \right] \quad (31)$$

■ 8. Bremsstrahlung cross-section

The expression of Bremsstrahlung cross-section differential in y , the photon energy k normalized to particle energy E , is [6]

$$d\sigma_{\text{Brems}} = 4 \alpha^3 \left(Z \frac{h c}{E_0} \right)^2 \left(\frac{4}{3} - \frac{4y}{3} + y^2 \right) \left[\text{Log} \left(2 \gamma \left(\frac{1}{y} - 1 \right) \right) - \frac{1}{2} \right] \frac{dy}{y} \quad (32)$$

where E_0 is the rest energy of the incident particle. As for elastic scattering, the total cross-section is defined by a minimum value of the variable, here the photon energy, which, in a storage ring, is related to the momentum acceptance of the machine. The above formula turns out to be valid for all the processes we have envisaged and is Lorentz invariant [7] so that the different cases of section 7 have not to be distinguished.

(* Bremsstrahlung cross-section *)

```
Brems [ym_, g1_, e0_, z_] = Block[{alpha, f, g, h, y, yM, t},
```

```
  (* differential cross-section *)
```

```
  f = (1 - 2 (1 - y) / 3 + (1 - y)^2) (Log[2 g1] + Log[1 - y] - Log[y] - 1/2) / y;
```

```
  (* total cross-section *)
```

```
  Unprotect [Log];
```

```
  rLog = Log[-1 + x_] -> Log[1 - x];
```

```
  Protect [Log];
```

```
  g[y_] = Integrate[f, y] /. rLog;
```

```
  h = g[yM] - g[ym];
```

```
  h = Limit[h, yM -> 1] + 4 PolyLog[2, ym] - 4 ym;
```

```
  h = Expand[h];
```

```
  h = Normal[Series[h ym, {ym, 0, 1}]] / ym;
```

```
  alpha = 1/137;
```

```
  r2 = .389/e0^2; (* [ mb ] *)
```

```
  h 4 alpha^3 r2 z^2]
```

or

$$\sigma_{\text{Brems}} = \frac{4 \alpha^3}{3} \left(Z \frac{h c}{E_0} \right)^2 \left(2 \pi^2 - \frac{1}{4} - \frac{5}{2} \text{Log} 2\gamma + 2 \text{Log} y_m (1 + \text{Log} y_m - 2 \text{Log} 2\gamma) \right) \quad (34)$$

■ References

- [1] S. Wolfram, *Mathematica*, Addison-Wesley (1988)
- [2] J.D. Jackson, *Classical Electrodynamics*, John Wiley & Sons (1975)
- [3] R. Aleksan et al., *Measuring CP violation in the B Bar system with asymmetric energy e+ e- beams*. SLAC-PUB 4673 (1988).
P. Grosse Wiesmann, *Colliding a linear electron beam with a storage ring beam*. SLAC PUB 4545 (1988).
- [4] Particle Data Group, *Review of Particle Properties*, Physics Letters B204 (1988).
- [5] N. Kroll, *Proceedings of the Workshop on High Luminosity Asymmetric Storage Rings* edited by D.G. Hitlin and F.C. Porter (California Institute of Technology, 1989).
- [6] J.M. Jauch and F. Rohrlich, *The Theory of Photons and Electrons*, Springer - Verlag (1976)
F. Porter, private communication.
- [7] Y.S. Tsai, *Pair Production and Bremsstrahlung of charged leptons*, Review of Modern Physics, Vol. 46, No 4 (1974).
V. Votruba, Physics Review **73**, 1468 (1948); Acad. Tcheque Sci. Bull. Intern. **49**, 19.

Index

Nota : The commands are listed in alphabetic order. They can be classified into two categories: graphs and functions.

For the graphs, the arguments must be given in the form of numerical values.

For the functions, the arguments may be given in either symbolic or numeric form. To get the numerical result, type

N[function[x, a, b, ...]]

Example:

N[Bhabha[Cos[.01], 100]

To plot a function, type

Plot[function[x, a, b, ...], {x, xmin, xmax}]

or

Plot3D[function[x, y, a, ...], {x, xmin, xmax}, {y, ymin, ymax}]

The italic arguments in the above statements must be given in the form of numerical values.

Examples:

Plot[energy2[e1, Pi/4, 0, 0], {e1, .1, 1}]

Plot3D[energy2[e1, t, 0, 0], {e1, .1, 1}, {t, 0, Pi/4}]

Bhabha[xm, s]

Bhabha cross-section [mbarn] in the center of mass frame for a value **xm** of the cosine of the minimum scattering angle χ_m defined in the center of mass frame and the square of the collision energy **s** [GeV**2].

Bhabha1[xm, s]

Term in $1 / (1 - xm)$ in the Bhabha cross-section series expansion .

Bhabhaa[e, m, s, chi0]

Asymmetric case. Bhabha cross-section [mbarn] for particles of energy **e** and mass **m** both normalized to the square of the collision energy **s** [GeV**2], and a minimum scattering angle **chi0** defined in the laboratory frame.

Bhabhas [t, m, s, chi0]

Symmetric case. Bhabha cross-section [mbarn] for beams crossing at angle $2t$. The particle mass m is normalized to the square of the collision energy s [GeV**2]. The minimum scattering angle χ_0 is defined in the laboratory frame.

Brems [ym, g, e0, z]

Bremsstrahlung cross-section [mbarn] in the laboratory frame for a value ym of the minimum photon energy normalized to the energy E of the incident particle, a ratio γ (g) of E to e_0 [GeV], the rest energy of the incident particle and a product z of the incident and target particles number of charges.

Coulomb [xm, e, z]

Coulomb cross-section [mbarn] in the laboratory frame for a value xm of the cosine of the minimum scattering angle χ_m , an energy e [GeV] of the incident particle and a product z of the incident and target particles number of charges.

energy2 [e1, t, m1, m2]

Asymmetric case. Beam 2 energy normalized to \sqrt{s} as a function of beam 1 energy E_1/\sqrt{s} (e_1), half-crossing angle t , particle mass in beam1 m_1/\sqrt{s} (m_1) and particle mass in beam 2 m_2/\sqrt{s} (m_2).

energyS [t, m1, m2]

Symmetric case. Square of beam energy normalized to \sqrt{s} as a function of half-crossing angle t , particle mass in beam1 m_1/\sqrt{s} (m_1) and particle mass in beam 2 m_2/\sqrt{s} (m_2).

graphS [space [v1], space [v2]]

3-D plot of the space part of the 4-vectors v_1 and v_2 .

graphVa [e1, t, m1, m2]

Asymmetric case. Plot of the momentum vectors of beam 1 (p_1), beam 2 (p_2) and center of mass (pcm) for given values of E_1/\sqrt{s} (e_1), half crossing angle t , m_1/\sqrt{s} (m_1), m_2/\sqrt{s} (m_2).

graphVs [t, m1, m2]

Symmetric case. Plot of the momentum vectors of beam 1 (p_1), beam 2 (p_2) and center of mass (pcm) for given values of half crossing angle t , m_1/\sqrt{s} (m_1), m_2/\sqrt{s} (m_2).

Lorentz [bg, phi, psi]

Matrix of a Lorentz transformation defined by $\beta\gamma$ (bg), ϕ (phi), ψ (psi).

Relativistic Kinematics

LorentzProd[{**bg1**, **phi1**, **psi1**}, {(**bg**)², **phi2**, **psi2**}]

Matrix of the product of two Lorentz transformations defined by $\beta\gamma_1$ (**bg1**), ϕ_1 (**phi1**), ψ_1 (**psi1**) and $\beta\gamma_2$ (**bg2**), ϕ_2 (**phi2**), ψ_2 (**psi2**).

LorentzV[{{**bg1**, **phi1**, **psi1**}, ... , {**bgn**, **phin**, **psin**}}, **v**]

Transform of a 4-vector **v** through the *n* Lorentz transformations defined by $\beta\gamma_1$ (**bg1**), ϕ_1 (**phi1**), ψ_1 (**psi1**) and $\beta\gamma_n$ (**bgn**), ϕ_n (**phin**), ψ_n (**psin**).

Moller[**xm**, **s**]

Moller cross-section [mbarn] in the center of mass frame for a value **xm** of the cosine of the minimum scattering angle χ_m and the square of the collision energy **s** [GeV**2].

Mollera[**e1**, **s**, **chi0**, **m1**, **m2**]

Asymmetric case. Moller cross-section [mbarn] in the center of mass frame for a beam 1 energy **e1**, a square energy **s** [GeV**2] in the center of mass frame, a minimum scattering angle **chi0** defined in the laboratory frame, a particle mass **m1** in beam 1 and a particle mass **m2** in beam 2.

Mollers[**t**, **s**, **chi0**, **m1**, **m2**]

Symmetric case. Moller cross-section [mbarn] in the center of mass frame for a half crossing angle **t**, a square energy **s** [GeV**2] in the center of mass frame, a minimum scattering angle **chi0** defined in the laboratory frame, a particle mass **m1** in beam 1 and a particle mass **m2** in beam 2.

momentum[**e**, **m**]

Momentum of a particle of energy **e** and mass **m**.

p1Va[**e1**, **t**, **m1**, **m2**]

Asymmetric case. Beam 1 momentum vector for given values of E_1/\sqrt{s} (**e1**), half crossing angle **t**, m_1/\sqrt{s} (**m1**), m_2/\sqrt{s} (**m2**).

p1Vs[**t**, **m1**, **m2**]

Symmetric case. Beam 1 momentum vector for given values of half crossing angle **t**, m_1/\sqrt{s} (**m1**), m_2/\sqrt{s} (**m2**).

p2Va[**e1**, **t**, **m1**, **m2**]

Asymmetric case. Beam 2 momentum vector for given values of E_1/\sqrt{s} (**e1**), half crossing angle **t**, m_1/\sqrt{s} (**m1**), m_2/\sqrt{s} (**m2**).

p2Vs [t, m1, m2]

Symmetric case. Beam 2 momentum vector for given values of half crossing angle t , $m1/\sqrt{s}$ ($m1$), $m2/\sqrt{s}$ ($m2$).

pcmMa [e1, t, m1, m2]

Asymmetric case. Center of mass momentum for given values of $E1/\sqrt{s}$ ($e1$), half crossing angle t , $m1/\sqrt{s}$ ($m1$), $m2/\sqrt{s}$ ($m2$).

pcmMs [t, m1, m2]

Symmetric case. Center of mass momentum for given values of half crossing angle t , $m1/\sqrt{s}$ ($m1$), $m2/\sqrt{s}$ ($m2$).

pcmVa [e1, t, m1, m2]

Asymmetric case. Center of mass momentum vector for given values of $E1/\sqrt{s}$ ($e1$), half crossing angle t , $m1/\sqrt{s}$ ($m1$), $m2/\sqrt{s}$ ($m2$).

pcmVs [t, m1, m2]

Symmetric case. Center of mass momentum vector for given values of half crossing angle t , $m1/\sqrt{s}$ ($m1$), $m2/\sqrt{s}$ ($m2$).

space [v]

Space part of a 4-vector v .

time [v]

Time component of a 4-vector v .

LAWRENCE BERKELEY LABORATORY
TECHNICAL INFORMATION DEPARTMENT
1 CYCLOTRON ROAD
BERKELEY, CALIFORNIA 94720